



Universidad de San Carlos de Guatemala  
Facultad de ingeniería  
Aux: Fabian Reyna  
Organización de lenguajes y compilares 1

Oscar David Padilla Vásquez  
202103250

## GRAMATICA BNF

**Terminales** = {CADENA, ENTERO, DECIMAL, ID, CARAC, PTCOMA, PRINT, MAS, MENOS, UMENOS, MULT, DIV, POTENCIA, MODULO, LLAVIZQ, LLAVDER, CORIZQ, CORDER, PARIZQ, PARDER, DOSPT, INCRE, DECRE, IGUAL, IGUALA, DIFER, MENOR, MENORQUE, MAYOR, MAYORQUE, OR, AND, XOR, NOT, FLECHA, DEFAULT, VAR, INT, RDOUBLE, BOOL, CHAR, RSTRING, CONST, IF, ELSE, MATCH, WHILE, FOR, DO, RTRUE, RFALSE, BREAK, CONTINUE}

**No terminales** = {INICIO, INSTRUCCIONES, INSTRUCCION, IMPRESION, DECLARACION, ASIGNACION, SIF, SWHILE, SDOWHILE, TIPOS, EXPRESION, CFOR, ACTU, INCREMENTO, DECREMENTO, BK, CT, SMATCH, LIST\_CASES, CASE\_MATCH}

### Producciones:

<INICIO> ::= <INSTRUCCIONES>

<INSTRUCCIONES> ::= <INSTRUCCIONES> <INSTRUCCION>  
| <INSTRUCCION>

<INSTRUCCION> ::= <IMPRESION>  
| <DECLARACION>  
| <ASIGNACION>  
| <SIF>  
| <CFOR>  
| <SWHILE>  
| <SMATCH>  
| <SDOWHILE>  
| <INCREMENTO> PTCOMA  
| <DECREMENTO> PTCOMA  
| error PTCOMA

<SMATCH> ::= MATCH <EXPRESION> LLAVIZQ <LIST\_CASES> LLAVDER

<LIST\_CASES> ::= <LIST\_CASES> <CASE\_MATCH>  
| <CASE\_MATCH>

<CASE\_MATCH> ::= <EXPRESION> FLECHA LLAVIZQ <INSTRUCCIONES> LLAVDER  
| DEFAULT FLECHA LLAVIZQ <INSTRUCCIONES> LLAVDER

<IMPRESION> ::= PRINT PARIZQ <EXPRESION> PARDER PTCOMA

<DECLARACION> ::= VAR ID DOSPT <TIPOS> PTCOMA  
| VAR ID DOSPT <TIPOS> IGUAL <EXPRESION> PTCOMA  
| CONST ID DOSPT <TIPOS> PTCOMA  
| CONST ID DOSPT <TIPOS> IGUAL <EXPRESION> PTCOMA

<CFOR> ::= FOR PARIZQ <ASIGNACION> <EXPRESION> PTCOMA <ACTU> PARDER LLAVIZQ  
<INSTRUCCIONES> LLAVDER

<ASIGNACION> ::= ID IGUAL <EXPRESION> PTCOMA

<SIF> ::= IF PARIZQ <EXPRESION> PARDER LLAVIZQ <INSTRUCCIONES> LLAVDER  
| IF PARIZQ <EXPRESION> PARDER LLAVIZQ <INSTRUCCIONES> LLAVDER ELSE LLAVIZQ  
<INSTRUCCIONES> LLAVDER  
| IF PARIZQ <EXPRESION> PARDER LLAVIZQ <INSTRUCCIONES> LLAVDER ELSE <SIF>

<SWHILE> ::= WHILE PARIZQ <EXPRESION> PARDER LLAVIZQ <INSTRUCCIONES> LLAVDER

<SDOWHILE> ::= DO LLAVIZQ <INSTRUCCIONES> LLAVDER WHILE PARIZQ <EXPRESION> PARDER  
PTCOMA

<INCREMENTO> ::= ID INCRE

<DECREMENTO> ::= ID DECRE

<ACTU> ::= ID IGUAL <EXPRESION>  
| <INCREMENTO>  
| <DECREMENTO>

<BK> ::= BREAK PTCOMA

<CT> ::= CONTINUE PTCOMA

<TIPOS> ::= INT  
| RDOUBLE  
| BOOL  
| CHAR  
| RSTRING

<EXPRESION> ::= MENOS <EXPRESION>  
| <EXPRESION> MAS <EXPRESION>  
| <EXPRESION> MENOS <EXPRESION>  
| <EXPRESION> MULT <EXPRESION>  
| <EXPRESION> DIV <EXPRESION>  
| <EXPRESION> POTENCIA <EXPRESION>  
| <EXPRESION> MODULO <EXPRESION>  
| <EXPRESION> IGUALA <EXPRESION>  
| <EXPRESION> DIFER <EXPRESION>  
| <EXPRESION> MENOR <EXPRESION>  
| <EXPRESION> MENORQUE <EXPRESION>  
| <EXPRESION> MAYOR <EXPRESION>  
| <EXPRESION> MAYORQUE <EXPRESION>  
| <EXPRESION> AND <EXPRESION>  
| <EXPRESION> OR <EXPRESION>  
| <EXPRESION> XOR <EXPRESION>  
| NOT <EXPRESION>  
| ENTERO  
| DECIMAL  
| CADENA  
| ID  
| CARAC  
| RTRUE  
| RFALSE  
| PARIZQ <EXPRESION> PARDER  
| PARIZQ <TIPOS> PARDER <EXPRESION>